
Credits

Credits:
Many slides (all the figures with blue background and few others)
were taken from Holzmann’s slides on “Logical Model Checking”,
course given at Caltech
(http://spinroot.com/spin/Doc/course/index.html)

Outline

1 Overview of Promela
Rules for Executability
Control Flow

2 Promela Semantics
Motivation
Operational Model
Interpreting Promela Models
Something on Verification

On Promela Expressions

Depending on the system state each statement in a Spin model
can be

Executable
Blocked

Any expression in Promela can be used as statements in any
context
Expressions are executable iff they evaluate to true or to a
non-zero integer value
For instance, instead of writing a waiting loop like:
while (a != b)

skip /* do nothing and wait till a==b */

it is possible to write in Promela
(a==b);

On Promela Expressions

Expressions must be side-effect free
Reason: a blocking expression may be evaluated many times
Exception to this rule: expressions containing the run operator can
have side effects

Syntactic restriction: There can be only one run operator in an
expression and it cannot be combined with any other operator

run must be used carefully!
active proctype splurge(int n)
{

printf("%d\n", n);
run splurge(n+1)

}

After the 255th attempt to instantiate a new process, Promela will
fail
The run expression will evaluate to zero and it will permanently
block the process
Since the process will not reach the end of its code, then will not
terminate (nor die) and none of its predecessor could die either

Rules for Executability
6 Types of Basic Statements

Always executable:
assignment: x++, x--, x = x+1, x = run P()

b = c++ is not a valid expression (right-hand side is not side-effect
free)

print: printf(‘‘x = %d\n’’, x)
assertion: assert(1+1==2)

Executable when true (non-zero):
expression statement:
(x), (1), run P(), skip, true, else, timeout

Executable when target channel is non-full
send: q!ack(m)

Executable when target channel is non-empty, and constraints are
met

receive: q?ack(n)

Defining the Control Flow

Basic statements (so far)
print, assignment, assertions, expressions, send and receive
Notice that run is not a statement but an operator and skip is an
expression (equivalent to (1) or true)

Five ways to define control flow
1 Semi-colons, gotos and labels
2 Structuring aids

inlines
macros

3 Atomic sequences (indivisible sequences)
atomic {...}
d_step {...}

4 Non-deterministic selection and iteration
if ... fi
do ... od

5 Escape sequences (for error handling/interruptions)
{...} unless {...}

Selection

The (non-deterministic) if statement is inspired on Dijkstra’s guarded
command language

the else guard is executable iff none
of the other guards is executable.

/* pick a number 0..3 */
if
:: n=0
:: n=1
:: n=2
:: n=3
fi

non-deterministically assigns

a value to n in the range 0..3

if
:: (n % 2 != 0)  -> n = 1
:: (n >= 0)      -> n = n-2
:: (n % 3 == 0)  -> n = 3
:: else /* -> skip */
fi

underlying

non-deterministic

automaton

/* find the max of x and y */
if
:: x >= y -> m = x
:: x <= y -> m = y
fi

Gerardo Schneider (Ifi, UiO) INF5140 - Lecture 8: Promela Semantics 13.03.2007 10 / 50



university-logo

Selection

else is a predefined variable

where in C one writes:

if (x <= y)

x = y-x; 

y++;

i.e., omitting the ‘else’

in Promela this is written:

if

:: (x <= y) -> x = y-x

:: else

fi;

y++

i.e., the ‘else’ part cannot be omitted

x <= y else

x = y-x

y++

in this case ‘else’ evaluates to:

!(x <= y)

the else clause always has to

be explicitly present

without it, the if- statement would

block until (x<=y) becomes true

(it then gives only one option for behavior)

no need to add
“-> skip”

Selection

timeout is also a predefined variable

if
:: q?msg -> ...
:: q?ack -> ...
:: q?err -> ...
:: timeout -> ...
fi

wait until an expected message

arrives, or recover when the system

as a whole gets stuck (e.g., due to

message loss)

note carefully that using‘else’

intead of ‘timeout’ is dubious

in this context

checking for bad timeouts:

spin –Dtimeout=true model

Selection

else and timeout are related
Both are predefined Boolean variables
Their values are set to true or false by the system, depending on
the context

They are, however, not interchangeable
else is true iff no other statement in the process is executable
timeout is true iff no other statement in the system is executable

A timeout may be seen as a system level else
Are these equivalent?

if
:: q?msg -> ...
:: q?ack -> ...
:: timeout -> ...
fi

if
:: q?msg -> ...
:: q?ack -> ...
:: else -> ...
fi

No! In the second, if a message is not received when the control
is at the if then the else is taken immediately

Repetition

The do statement is an if statement caught in a cycle

do
:: guard1 -> stmnt1.1; stmnt1.2; stmnt1.3;...
:: guard2 -> stmnt2.1; stmnt2.2; stmnt2.3;...
::...
:: guardn -> stmntn.1; stmntn.2; stmntn.3;...
od

Only a break or a goto can exit from a do

A break transfers control to the end of the loop

Repetition

There are many ways of writing a waiting loop, by exploiting the
executability rules it’s possible to simplify the model

do

:: (a == b) -> break

:: else -> skip

od

L: if

:: (a==b) -> skip

:: else -> goto L

fi

the skip is not needed here

and can introduce an

unnecessary control state

(a == b)

these two constructs

are equivalent to a single

expression statement

else

a==b

a==b else

skip

a==b

note that ‘break’,
like ‘goto’, is not
a basic statement
but a control-flow
specifier

Atomic Sequences

atomic { guard -> stmnt1; stmnt2; ... stmntn }

– executable if the guard statement is executable

– any statement can serve as the guard statement

– executes all statements in the sequence without

interleaving with statements in other processes

– if any statement other than the guard blocks, atomicity is lost 

atomicity can be regained when the statement becomes 

executable

– example: mutual exclusion with an indivisible test&set:

active [10] proctype P()
{ atomic { (busy == false) -> busy = true };
mutex++;

assert(mutex==1);

mutex--;
busy = false;

}

Deterministic Steps

d_steps are more restrictive and more efficient than atomic
sequences

d_step { guard -> stmnt1; stmnt2; ... stmntn }

– like an atomic, but must be deterministic and may not block 

anywhere

– especially useful to perform 

intermediate computations

with a deterministic result,

in a single indivisible step

– atomic and d_step sequences are often used as a model
reduction method, to lower complexity of large models
(improving tractability)

d_step { /* reset array elements to 0 */
i = 0;
do
:: i < N -> x[i] = 0; i++
:: else -> break
od;
i = 0

}

Atomic Sequences, Deterministic Steps and Gotos

• goto-jumps into and out of atomic sequences are 

allowed

– atomicity is preserved only if the jump starts inside on atomic 

sequence and ends inside another atomic sequence, and 

the target statement is executable

• goto-jumps into and out of d_step sequences are 

forbidden
d_step {

i = 0;
do
:: i < N -> x[i] = 0; i++
:: else -> break
od

};
x[0] = x[1] + x[2];

this is a jump out
of the d_step sequence
and it will trigger an
error from Spin

the problem is prevented in this
case by adding a “; skip” after the
od keyword – there’s no runtime penalty for
this, since it’s inside the d_step

Deterministic Steps vs Atomic Sequences

Both sequences are executable only when the first (guard)
statement is executable

atomic: if any other statement blocks, atomicity is lost at that point;
it can be regained once the statement becomes executable later
d_step: it is an error if any statement other than the (first) guard
statement blocks

Other differences:
d_step: the entire sequence is executed as one single transition
atomic: the sequence is executed step-by-step, but without
interleaving, it can make non-deterministic choices

Remarks
Infinite loops inside atomic or d_step sequences are not detected
The execution of this type of sequence models an indivisible step,
which means that it cannot be infinite

Deterministic Steps and Atomic Sequences

execution with

full interleaving

active proctype P1() { t1a; t1b }
active proctype P2() { t2a; t2b }

Deterministic Steps and Atomic Sequences
execution with one

atomic sequence

active proctype P1() { atomic { t1a; t1b } }
active proctype P2() { t2a; t2b }

P1 could make alternate choices at

the intermediate states (e.g., in if

or do-statements)

P2 can be interrupted, but not P1

Deterministic Steps and Atomic Sequences
execution with a

d_step sequence

active proctype P1() { d_step {t1a; t1b} }
active proctype P2() { t2a; t2b }

no intermediate states are created: 

faster, smaller graph, but no non-

determinism possible inside d_step

sequence itself

P1 now has only one transition…

(0,0)

(0,1)

(1,0)

(1,1)

(-,1)

(-,0)
(0,2)

(1,2)

(-,2)

(0,-)

(1,-)

(-,-)

end

end

end

end
end

t1a;t1b

t1a;t1b

t1a;t1b

t1a;t1b

t2a

t2b

end

end

t2a

t2bt2a

t2b

0

1

t1a;t1b

end

P1

0

1

2

t2a

t2b

end

P2

Escape Sequences

Syntax: { P } unless { Q }

Execution starts with the statements from P

Before executing each statement in P, the executability of the first
statement in Q is checked
Execution of P statements continue only if the first instruction of Q
is not executable
As soon as the Q first statement can be executed, then control
changes and execution continues in Q
Example
A; { do

:: b1 -> B1
:: b1 -> B1
...
od } unless { c -> C };

D

c acts here as a watchdog: as soon as it becomes true, C is
executed and then D

Inline definitions

• somewhere in between a macro and a procedure

• used as replacement text with textual name substitution through 

parameters (it is a named piece of text with optional parameters)

• an inline is not a function – it cannot return values to the caller

• can help to structure a model

• compare:

#define swap(a,b) tmp = a; \

a = b; \

b = tmp

#define swap(a,b) tmp = a; \

a = b; \

b = tmp

inline swap(a,b) {

tmp = a;

a = b;

b = tmp

}

inline swap(a,b) {

tmp = a;

a = b;

b = tmp

}

looks a little cleaner

line nr refs are better

hint:
when confused, usespin –I spec.pml

to show the result of all inlining

and macro preprocessing operations...

Gerardo Schneider (Ifi, UiO) INF5140 - Lecture 8: Promela Semantics 13.03.2007 24 / 50



university-logo

Automata and “proctypes”

Each Promela proctype defines a finite state automaton
(S, s0, L, T , F )

The set of states S corresponds to the possible points of control
within proctype
The transition relation T defines the flow of control
The labels in L relates each transition in T with a basic statement
that defines the executability and the effect of that transition
The set of final states F is defined with Promela end-states,
accept-states and progress-states1

There are only 6 basic statements: assignments, assertions, print,
send and receive statements and expression statements
Anything else (if, goto, do, break, unless, atomic, d_step)
serves only to specify control flow and cannot appear as labels
on transitions
Every basic statement (including expressions) has a precondition
defining when it can be executable and its effect

1More on such states on next lecture

Automata and “proctypes”
Example

active proctype not_euclid()
{
S: if

:: x == y ->
assert(x != y);
goto L

:: x > y ->
L: x = x - y

:: x < y ->
y = y - x

fi;
E: printf(‘‘%d\n’’, x)
}

The corresponding automaton:

L:

x==y x<yx>y

assert

x=x-y y=y-x

printf(“%d\n”,x)

stop

start

S:

E:

Motivation

Each Promela proctype defines a finite state automaton
(S, s0, L, T , F )

The specification of a collection of asynchronous processes may be
written in Promela as the asynchronous product of the automata

We can obtain a (possible huge) graph (automaton) containing all
the reachable states where each edge represent a single possible
execution step
The structure of such graph (automaton) is determined by the
semantics of Promela

Without such semantics we cannot know which execution paths are
possible

We will see in a future lecture, how to specify properties in
Promela which will express claims about presence or absence of
subgraphs or paths in the reachability graph

Motivation
Example

chan x = [0] of { bit };
chan y = [0] of { bit };

active proctype A()
{

x!0 unless y!0
}

active proctype B()
{

y?0 unless x?0
}

x!0 y!0

A

stop

y?0 x?0

B

stop

What are the possible executions?
There are two possible handshakes

y!0 with y?0
x!0 with x?0

Are both handshakes possible?
We will find answers to these kind of questions later

Introduction

The operational model is based on the specification of a semantic
engine which determines how a Promela model defines system
executions, including the rules that apply to interleave execution of
process actions
The semantic engine operates on the global reachability graph

Such graph contains abstract objects corresponding to
asynchronous processes, variables and message channels

It only deals with local states and transitions
It does not know anything about control-flow construct (as if, do,
break and goto)

The global reachability graph determines
A global system state (nodes), defined in terms of:

Variables, messages, message channels and processes
A state transition (edges), defined in terms of:

The basic statements labeling the transitions
Transition selection and transition execution

Abstract Objects

We will see formal definitions of the following abstract objects (in which
the semantic engine operates on)

Variables
Messages
Channels
Processes
Transitions
System states

Note: Formal definitions of basic terms like sets, indentifiers, integers and booleans
are not presented

Variables

• a promela variable is defined by a five-tuple

{name,scope,domain,inival,curval}

short x=2, y=1;

active proctype not_euclid()
{
S: if

:: x > y  -> L: x = x - y
:: x < y  ->    y = y - x
:: x == y ->    assert(x != y); goto L
fi;

E: printf(“%d\n”, x)
}

name

scope: global
inival x: 2domain: -215..215-1

curval of x at E: 1

curval of x at S: 2

Messages

• a message is a finite, ordered set of variables

(messages are stored in channels – defined next)

mtype = { req, resp, ack };

chan q = [2] of { mtype, bit };

active proctype not_very_useful()
{  bit p;

do
:: q?req,p -> q!resp,p
:: q?resp,p -> q!ack,1-p
:: q?ack,_
:: timeout -> break
od

}

parallel value assignment

domains:
mtype
bit

place names for values held in
message channel:

slot1.field1
slot1.field2

Channels

• a message channel is defined by a 3-tuple

{ch_id, nslots, contents}

chan q = [2] of { mtype, bit };

{
{slot1.field1, slot1.field2},
{slot2.field1, slot2.field2}

}

an ordered set of messages
maximally with nslots elements:

a ch_id is an integer 1..MAXQ that can be
stored in a variable

(ch_id’s <= 0 or > MAXQ do not correspond
to any instantiated channel, so the default
initial value of a variable of 0 is not a
valid ch_id)

variables of type chan

are either local or global,

but channels always

have global scope

(so, ch_id’s are always meaningful

when passed from one process

to another)

Channels
Channel Scope

chan r = [0] of { chan };

active proctype A()
{

chan q = [0] of { int };

r!q;
q?100

}

active proctype B()
{

chan s;

r?s;
s!100

}

global variable r
holds the ch_id of C1

C1 never disappears

local variable q in A
holds the ch_id of C2
(a global object)

C2 is created when A()
is instantiated
it disappears when A()
disappears

local variable s in B
is initialized to 0
(not a valid ch_id)

it is set to C2 in the
receive from r == C1

C2 is a globally visible
object with a limited
lifetime...

global objects

local
objects

r

q

C1

C2

s

in the initial system state
r, q, s, C1, and C2 all exist
r points to C1, q points to C2

Processes

• a process is defined by a six-tuple

{pid, lvars, lstates, inistate, curstate, transitions}

process
instantiation
number

finite set of
local variables

a finite set of integers
defining local proc states

the initial state a finite set of
transitions (to be defined)
between elements of lstates

the current state

process p
p.curstate
p.pid
etc.

Transitions

• a transition is defined by a seven-tuple

{tri_id, source-state, target-state, cond, effect, priority, rv}

x<y

(x<y)

nil

predefined system variables that are
used to define the semantics of
unless and rendezvous

transition t:
t.source
t.target
t.cond
t.effect
etc.

condition and effect are
defined for each basic
statement, and they are
typically defined on
variable and channel
values, possibly also on
process states

System States

• a global state is defined by a eight-tuple

{ gvars, procs, chans, exclusive, handshake, timeout, else, stutter}

a finite set of
global variables

a finite set of
processes

a finite set of
message channels

predefined integer system variables
that are used to define the
semantics of atomic, d_step, and
rendezvous

predefined Boolean
system variables

for stutter
extension rule

the global system state is called
the system “state vector”

The Semantics Engine

The semantics engine executes a Spin model step-by-step
In each step, one executable basic statement is selected
To determine whether a given statement is executable, the
executable clause –specified in the Promela manual– must be
evaluated
If more than one statement is executable, one of them is
non-deterministically chosen

Correctness of Spin models is independent of the selection criterion
For the selected statement, the effect clause (specified in the
Promela manual) is applied
The control state (curstate) of the process that executes the
statement is updated
The semantics engine continues executing statements until no
executable statements remain

There is no more processes to execute; or
The remaining processes are deadlocked

The Semantics Engine
The Algorithm

global states s, s’
processes     p, p’
transitions   t, t’

1 while ((E = executable(s)) != {}) 
2 { 
3      for some (process p and transition t) from E
4      {    s' = apply(t.effect, s)
5
6           if (handshake == 0)
7           {    s = s'
8                p.curstate = t.target
9           } else 

10           { /* try to complete rv handshake */
11                E' = executable(s') 
12 /* if E' is {}, s is unchanged  */
13
14                for some (process p’ and transition t') from E' 
15                {    s = apply(t'.effect, s') 
16                     p.curstate = t.target
17                     p'.curstate = t'.target
18 handshake = 0 
19 break

20      }     }   }
21 } 
22 while (stutter) { s = s } /* stutter extension rule */

Examples
Rendez-vous Handshake

Example 1: Priority for sending in both processes

chan x = [0] of { bit };
chan y = [0] of { bit };

active proctype A()
{

x?0 unless y!0
}

active proctype B()
{

y?0 unless x!0
}

Q: what is the combined
system behavior?

A: a non-deterministic
selection between

x!0;x?0
and

y!0;y?0

end

x!0 y!0

y?0x?0

handshake=1 handshake=2

end

x?0 y!0

end

y?0 x!0

unless escapes have
higher priority

Examples
Rendez-vous Handshake

Example 2: Priority for sending in one process anf receiving in the
other

chan x = [0] of { bit };
chan y = [0] of { bit };

active proctype A()
{

x!0 unless y!0
}

active proctype B()
{

y?0 unless x?0
}

Q: what is the combined
system behavior?

is it
x!0;x?0

or
y!0;y?0

?

end

x!0 y!0

end

y?0 x?0

unless escapes have
higher priority

end

y!0

y?0

handshake=2

A: only y!0;y?0 can happen

Examples
Rendez-vous Handshake

Example 3: Priority for receiving in both processes

chan x = [0] of { bit };
chan y = [0] of { bit };

active proctype A()
{

x!0 unless y?0
}

active proctype B()
{

y!0 unless x?0
}

Q: what is the combined
system behavior?

is it
x!0;x?0

or
y!0;y?0

?

end

x!0 y?0

end

y!0 x?0

unless escapes have
higher priority

end

x!0 y!0

y?0x?0

handshake=1 handshake=2

A: a non-deterministic
selection between

x!0;x?0
and

y!0;y?0

Examples
Comparing Example 1 and 3

chan x = [0] of { bit };
chan y = [0] of { bit };

active proctype A()
{

x!0 unless y?0
}

active proctype B()
{

y!0 unless x?0
}

end

x!0 y!0

y?0x?0

chan x = [0] of { bit };
chan y = [0] of { bit };

active proctype A()
{

x?0 unless y!0
}

active proctype B()
{

y?0 unless x!0
}

end

x!0 y!0

y?0x?0

same global behavior
but for very different
reasons....

Verification and Promela Semantics

Remarks
The semantics engine does not establish the validity or invalidity
of correctness requirements

The judgment of what is a correct system behavior is outside the
definition of the Promela semantics

The addition of a verification option does not affect the semantics
of a Promela model

The semantics engine has no interpretation of valid end states,
accepting states, non-progress states, never claims, trace
assertions, etc
These language elements have no formal semantics within the
model
Assertion statements, special labels, never claims, etc, are
meta-statements about the semantics of the model
How such meta-statements are to be interpreted is defined in a
verifier, as part of the verification algorithm

Verification and Promela Semantics
Few Simple Examples

Without entering into details, these are few examples of how the
verifier and the semantics engine are related

When a verifier checks for safety properties, the predefined
system variable stutter (used in the last line of the semantics
engine) is set to false
When a verifier checks for liveness properties, the predefined
system variable stutter is set to true
A never claim does not define new semantics, but is used to
identify which part of the existing semantics can violate an
independently stated correctness criterion

Only infinite executions that are consistent with the formal
semantics of the model and with the constraint expressed by the
never claim can be generated

These comments will become clear after next lecture (on the definition
of correctness claims)...

Further Reading

The first part of this talk was based on Chapter 3 of Holzmann’s
book “The Spin Model Checker” and the second part was based
on Chap. 7 of the same book
Next lecture we’ll see how to define Correctness Claims
(Holzmann’s book Chap. 4)

